
Cos’è PL/SQL?

 Il linguaggio procedurale per l’estensione del linguaggio SQL di 

proprietà di Oracle

 Lo standard SQL è esteso dai principali sistemi commerciali:

 Da Oracle con PL/SQL (e Java)

 Da Access con Visual Basic

 Da SQL Server con Transact-SQL

Procedurale vs dichiarativo?

 Il linguaggio SQL è un linguaggio dichiarativo ottimale per inserire e 

reperire dati da un database ma non sufficientemente potente per 

codificare la logica applicativa

 I programmi imperativi definiscono in modo esplicito un algoritmo per conseguire 

uno scopo

 I programmi dichiarativi definiscono in modo esplicito soltanto lo scopo da 

raggiungere, lasciando che l'implementazione dell'algoritmo sia realizzata dal 

software di supporto. Il DBMS appunto!

 Un esempio non implementabile in SQL

 Si vuole aumentare lo stipendio ai dipendenti che:

• Non abbiamo avuto più di 3 aumenti negli ultimi 5 anni

• Il cui rendimento è superiore alla mediana degli altri dipendenti dello stesso 

dipartimento. Il rendimento è calcolato in base a un insieme di formule 

matematiche da usare in alternativa in base alle caratteristiche dell’impiegato

 Nel caso in cui l’aumento sia applicabile

• Va aggiornato lo stipendio dell’impiegato

• Va compilato un report

• Va inviata una mail al direttore del dipartimento e all’impiegato

Programmazione nei DBMS

 Server-side Programming: la logica di programmazione risiede 

all’interno del database mediante linguaggi supportati dal DBMS. Nel 

caso di Oracle: PL/SQL e Java

 Client-side Programming: i comandi SQL sono embedded nelle 

applicazioni sviluppate con un linguaggio procedurale (es. C++, Java)

 Utilizzo di precompiler

 Utilizzo di API (es. JDBC, OCI)

 La programmazione server-side pone molti vantaggi rispetto a quella 

client-side

 Maggiori performance: minor quantità di dati trasferiti in rete

 Minore quantità di memoria richiesta: una sola copia della procedura è caricata 

nella shared memory

 Maggiore produttività: le procedure condivise da più applicazioni non devono 

essere replicate

 Sicurezza: gli accessi alla procedura e dati sono regolati dai permessi degli utenti 

del db e non degli utenti dell’applicaizone

Il motore di PL/SQL

 Esegue le porzioni procedurali del codice ma invia al server 

oracle i comandi SQL

 Un blocco deve essere compilato prima che possa essere 

eseguito

 Controllo sintattico

• Struttura del comando, parole riservate e variabili

 Binding

• Controlla che gli oggetti referenziati esistano

 Generazione del p-code

• Istruzioni che il motore PL/SQL può eseguire

Blocchi PL/SQL

 I blocchi PL/SQL (Block)

 Rappresentano l’unità elementare di codice PL/SQL

 Normalmente contengono i comandi sufficienti a eseguire uno specifico 

compito

 Esistono due tipi di blocchi PL/SQL

 Anonymous

 Named: Si tratta di blocchi PL/SQL precompilati che vengono memorizzati 

nel database

• stored procedure

• function

• trigger

• package: gruppi di procedure e funzioni assemblate assieme 

tipicamente per affinità funzionale

Struttura di un blocco PL/SQL
 Sezione di dichiarazione

 Per dichiarare, variabili, costanti, cursori,ecc.

 E’ opzionale

 Sezione di esecuzione

 Descrive la logica dei comandi

 Può contenere istruzioni SQL

 E’ obbligatoria

 Sezione di gestione delle eccezioni

 Viene eseguita quando si presentano degli errori

 E’ opzionale

Attenzione nella definizione delle procedure e 

funzioni la clausola DECLARE è implicita

DECLARE

BEGIN

EXCEPTION

END;

Procedure I
 Una procedura è un blocco di codice PL/SQL dotato di un 

nome che viene mantenuto all’interno del database (stored 

procedure)

 La clausola IS sostituisce la clausola DECLARE

CREATE OR REPLACE PROCEDURE nome_procedura 
[(parametri)] IS

Definizioni;

BEGIN

Corpo procedura;

END;

Procedure II

 Una procedura può essere richiamata utilizzando il comando 
call

 Parametri è una sequenza di

che specifica eventuali valori passati in input

 TIPO_DATO non deve specificare lunghezza, precisione o 
scala.

VARCHAR2(10) non è un tipo di dato valido VARCHAR2 si! 

 Oracle deriva lunghezza, precisione o scala degli argomenti 
dall’ambiente da cui la procedura è chiamata.

 Di default i parametri sono utilizzati solo per il passaggio in 
ingresso delle informazioni (IN). Il passaggio IN OUT equivale 
a un passaggio di dati per riferimento in C.

[IN |OUT|IN OUT] <nome parametro> <tipo parametro>[,]

CALL nome_procedura([parametri]);

Funzioni
 Le funzioni sono del tutto simili a procedure a meno della 

clausola RETURN che specifica il tipo di tato restituito

CREATE FUNCTION nome_funzione … RETURN BOOLEAN IS

Definizioni;

BEGIN

Corpo procedura;

RETURN Variabile;

END;

Esempio: anonymous

DECLARE

qty_on_hand NUMBER(5);

BEGIN

SELECT quantity INTO qty_on_hand

FROM inventory

WHERE product = 'TENNIS RACKET' FOR UPDATE OF quantity;

IF qty_on_hand > 0 THEN -- check quantity

UPDATE inventory SET quantity = quantity – 1

WHERE product = 'TENNIS RACKET';

INSERT INTO purchase_record

VALUES ('Tennis racket purchased', SYSDATE);

ELSE

INSERT INTO purchase_record

VALUES ('Out of tennis rackets', SYSDATE);

END IF;

COMMIT;

END;

Esempio1: named
CREATE OR REPLACE PROCEDURE Esempio IS

qty_on_hand NUMBER(5);

BEGIN

SELECT quantity INTO qty_on_hand

FROM inventory

WHERE product = 'TENNIS RACKET' FOR UPDATE OF quantity;

IF qty_on_hand > 0 THEN -- check quantity

UPDATE inventory SET quantity = quantity – 1

WHERE product = 'TENNIS RACKET';

INSERT INTO purchase_record

VALUES ('Tennis racket purchased', SYSDATE);

ELSE

INSERT INTO purchase_record

VALUES ('Out of tennis rackets', SYSDATE);

END IF;

COMMIT;

END;

Esempio2: named

CREATE OR REPLACE PROCEDURE WHOIS(v_writerid VARCHAR2) IS

v_writername VARCHAR2(50);

BEGIN

SELECT fn || ' ' || ln INTO v_writername

FROM writer

WHERE writerid = v_writerid;

DBMS_OUTPUT.PUT_LINE('Writer ' || v_writerid || ' is ' || v_writername);

EXCEPTION

WHEN NO_DATA_FOUND THEN

DBMS_OUTPUT.PUT_LINE('No such writer: '|| v_writerid);

END;

DBMS_OUTPUT.PUT_LINE

 Mostra l’output a video 

 La procedura scrive l’’output su un buffer dell’SGA da cui può 
essere letto mediante il comando .get_line

 In SQL Developer il risultato di un comando compare nella 
finestra Output DBMS

 DBMS_OUTPUT è un package

 .PUT_LINE è una procedure all’interno del package

Dichiarazione di una variabile PL/SQL

 La dichiarazione deve essere effettuata nella sezione DECLARE

 Per default le variabili sono inizializzate a NULL

 Le variabili sono dichiarate e inizializzate ogni volta che si 
accede al blocco

 Due variabili con lo stesso nome devono essere dichiarate in 
blocchi diversi

 Consigli:
 Naming Conventions

• Fino a 30 caratteri, non case sensitive, cominciano con una lettera e non possono 
contenere spazi

• Non definire una variabile con il nome della colonna se queste vengono usate 
contemporaneamente. Utilizzare per esempio un prefisso per distinguerle (es. 
Quantity e vQuantity)

 Dichiarate una variabile per riga

identifier [CONSTANT] datatype [NOT NULL]   

[:= | DEFAULT expr];

Assegnamento di un valore a una 

variabile

CREATE PROCEDURE Esempio IS

c_tax_rateCONSTANT NUMBER(3,2) := 8.25;

... 

BEGIN

... 

v_hiredate := '31-DEC-98';

v_fullname := ln || ', ' || fn;

... 

variablename := expression;

 Viene effettuato tramite comando di assegnamento nella
sezione di esecuzione

 … oppure tramite il comando SELECT INTO

CREATE PROCEDURE Esempio IS

v_max_len   number(7);

BEGIN

SELECT max(length) INTO v_max_len

FROM article;

... 

4 tipi di variabili

 Scalar
 Possono contenere un singolo valore

 Corrispondono ai tipi di dati previsti per le tabelle Oracle più poche altre 

(es: Boolean)

 Composite
 Permettono di manipolare gruppi di campi

 es: una variabile di tipo %ROWTYPE memorizza un’intera riga

 Reference
 Contengono puntatori

 LOB (Large OBjects)
 Contengono elementi, chiamati locators, che specificano la posizione di 

oggetti di grosse dimensioni (es. immagini) che sono memorizzati 

separatamente

I principali tipi di dati scalari I

 VARCHAR2 (lung. max.)
 Fino a 32,767 byte

 CHAR [(lung. max.)]
 Fino a 32,767 byte

 NUMBER [(precisione, scala)]
 precisione: 0-38

 scala: -84 to 127

 NUMBER(5,2) -> ddd.dd

 DATE
 Da: January 1, 4712 BC A: December 31, 9999 AD

 BOOLEAN
 TRUE  o  FALSE  o  NULL

 Non ha nessun tipo corrispondente nei tipi degli attributi

CREATE PROCEDURE Esempio IS

v_job VARCHAR2(9);

v_total_sal NUMBER(9,2) := 0;

v_duedate  DATE := SYSDATE + 7;

v_valid BOOLEAN NOT NULL := TRUE;

c_tax_rate CONSTANT NUMBER(3,2) := 8.25;

BEGIN………………

Le variabili %TYPE I

 Quando si definisce una variabile PL/SQL per memorizzare il 
valore di una colonna è necessario assicurarsi la 
corrispondenza tra i due tipi di dato
 In caso contrario si verificherà un errore PL/SQL durante l’esecuzione

 Un tipo di dato “anchored” evita questo problema

 Se cambia la definizione della colonna, cambia anche 
runtime la definizione della variabile. Si realizza così 
l’indipendenza dei dati e si permette ai programmi di 
adattarsi ai cambiamenti del database

 %TYPE dichiara una variabile in base a: 

 La definizione di una colonna del database

 Un’altra variabile definita precedentemente

 Possibili prefissi per %TYPE sono:

 I nomi della tabella e della colonna

 Il nome della variabile precedentemente definita

Le variabili %TYPE II

CREATE PROCEDURE Esempio IS

v_writerid writer.writerid%TYPE;

v_length      article.length%TYPE;

v_min_length v_length%TYPE := 0;

BEGIN

………

Lifetime

 Indica l’intervallo durante il quale una variabile esiste in 

memoria e può contenere un valore

 Lo spazio in memoria è allocato quando la variabile 

viene dichiarata

 Lo spazio in memoria è deallocato quando il programma 

raggiunge il comando END del blocco in cui è stata 

creata

Scope (Visibilità)

 La regione del programma in cui referenziare una 

variabile

 Le variabili dichiarate in una blocco PL/SQL sono locali al 

blocco e sono considerate globali per tutti i sotto blocchi

 La visibilità è inibita se nel blocco viene dichiarata una 

variabile con lo stesso nome. 

 Un blocco può fare riferimento a variabili dichiarate nei blocchi 

padre

 Un blocco NON può fare riferimento a variabili dichiarate nei 

blocchi figli

Esempio

DECLARE

v_sal NUMBER(7,2) := 60000;

v_comm NUMBER(7,2) := v_sal * .20;

v_message VARCHAR2(255) := ' eligible for commission';

BEGIN 

DECLARE

v_sal         NUMBER(7,2) := 50000;

v_comm  NUMBER(7,2) := 0;

v_total_comp  NUMBER(7,2) := v_sal + v_comm;

BEGIN 

v_message := 'CLERK not'|| v_message;

END;

v_message := 'SALESMAN'|| v_message;

END;

SELECT INTO
SELECT  select_list

INTO {variable_name[, variable_name]...

| record_name}

FROM table

WHERE condition;

 E’ necessario indicare ordinatamente il nome di una variabile per 

ogni colonna selezionata.

 L’interrogazione deve restituire una e una sola tupla

 In caso contrario si genererà un errore

 PL/SQL gestisce questi due errori generando due exception predefinite, che 

possono quindi essere trattate nella sezione EXCEPTION

• NO_DATA_FOUND

• TOO_MANY_ROWS

Esempio SELECT INTO

CREATE PROCEDURE Esempio(v_writerid writer.writerid%TYPE) IS

v_lastname  varchar2(90);

v_amount    number(9,2);

BEGIN

SELECT ln, amount

INTO  v_lastname, v_amount

FROM   writer

WHERE  writerid = v_writerid;

DBMS_OUTPUT.PUT_LINE('Writer: ' || v_lastname || ' earns 

$' || v_amount);

END;

SQL statico in PL/SQL

 Con il termine SQL statico si identificano i comandi SQL 
direttamente inclusi nel codice PL/SQL e sottoposti al 
processo di compilazione

 In SQL statico:
 DDL non è supportato

• CREATE TABLE, CREATE INDEX, ALTER TABLE, DROP VIEW

 DCL non è supportato

• GRANT, REVOKE, CREATE USER, DROP ROLE, ALTER USER

 DML è supportato

• INSERT, UPDATE, DELETE

 TCL è supportato 

• COMMIT, ROLLBACK, SAVEPOINTC

Esempio

DECLARE

v_sal_increase   emp.sal%TYPE := 2000;

BEGIN

UPDATE  emp

SET sal = sal + v_sal_increase

WHERE   job = 'ANALYST';

COMMIT;

END;

CREATE OR REPLACE PROCEDURE Esempio IS

v_deptno   emp.deptno%TYPE := 10;

BEGIN

DELETE FROM   emp

WHERE         deptno = v_deptno;

COMMIT;

END;

SQL Dinamico

 La creazione dinamica di comandi SQL all’interno di un 
blocco di codice PL/SQL può essere utile quando:
 Il comando SQL non è noto a compile time

 Il comando SQL non è supportato come SQL statico

• GRANT, REVOKE, CREATE USER, DROP ROLE, ALTER USER

 Dove l’SQL dinamico non è necessario, l’SQL statico è 
preferibile perché la compilazione verifica la correttezza 
sintattica del comando, e degli oggetti che esso 
referenzia

 Esistono più metodi per costruire un comando di SQL 
dinamico:
 EXECUTE IMMEDIATE

 OPEN FOR, FETCH, CLOSE

 DBMS_SQL Package

EXECUTE IMMEDIATE

EXECUTE IMMEDIATE [dynamic SQL string statement without 

terminator]

[INTO {define_variable [, define_variable] ... | record}]

[USING [IN|OUT|IN OUT] bind_argument [, [IN|OUT|IN OUT] 

bind_arguments] ]

 Il comando può non richiede parametri di input/output

 In questo caso INTO e USING non sono necessari

 La stringa SQL non deve terminare con ; 

 Se il comando è un SELECT che restituisce una sola tupla si può 

utilizzare

 INTO per specificare le variabili di output

 USING per specificare le variabili di input e output

 La stringa SQL può contenere placeholder per argomenti di binding, 

ma tali argomenti non possono essere utilizzati per passare i nomi 

degli oggetti dello schema (tabelle o colonne). Si possono passare 

interi, date e stringhe ma non booleani o valori nulli

 Negli altri casi è necessario comporre la stringa SQL

Esempio EXECUTE IMMEDIATE
create or replace function get_row_cnts( p_tname in varchar2 )

return number IS

l_cnt number;

begin

execute immediate 'select count(*) from ' || p_tname into l_cnt;

return l_cnt;

end;

CREATE OR REPLACE FUNCTION get_num_of_employees (p_loc VARCHAR2, 

p_job VARCHAR2) RETURN NUMBER IS 

v_query_str VARCHAR2(1000); 

v_num_of_employees NUMBER; 

BEGIN 

v_query_str := 'SELECT COUNT(*) FROM emp_' || p_loc || ' WHERE job 

= :bind_job'; 

EXECUTE IMMEDIATE v_query_str INTO v_num_of_employees USING p_job; 

RETURN v_num_of_employees; 

END;

DEBUGGING

 All’atto della creazione il motore di PL/SQL indica se 

una procedura o funzione è stata creata 

correttamente o con errori

 Dal log del compilatore è possibile elencare i singoli 

errori

Esercizi in aula
EX1: Scrivere una procedura che scriva in output la stringa ESAME DI SISTEMI 

INFORMATIVI come concatenazione di 4 variabili

EX2: Scrivere una procedura/funzione che, dato il codice fiscale di un fornitore 

restituisca il suo nome

Controllo del flusso di 

elaborazione

 Per cambiare il flusso di esecuzione all’interno di un blocco di 

codice sono disponibili i seguenti comandi

 IF-THEN
• Seleziona se eseguire o non un comando

 IF-THEN-ELSE
• Seleziona quale di due comandi debbano essere eseguiti in mutua esclusione

 IF-THEN-ELSIF
• Seleziona quale di più comandi debbano essere eseguiti in mutua esclusione

 Attenzione:

 ELSIF è una parola

 END IF sono due parole

IF-THEN

 Se la condizione è 

TRUE, allora viene 

eseguito il ramo THEN

 Se la condizione è 

FALSE o NULL, allora il 

ramo THEN  non viene 

eseguito

 In entrambi i casi il 

flusso viene ripreso al 

primo comando che 

segue END IF

IF condition

FALSETRUE

THEN actions

IF condition THEN

statement(s);

END IF;

Esempio

...

IF sales > quota THEN

bonus:=compute_bonus(empid);

UPDATE payroll

SET pay = pay + bonus

WHERE empno = emp_id;

END IF; 

...

IF-THEN-ELSE

IF condition

TRUE

THEN actions

FALSE

ELSE actions

 Se la condizione è TRUE, 

allora viene eseguito il 

ramo THEN

 Se la condizione è FALSE 

o NULL, allora viene 

eseguito il ramo ELSE  

 In entrambi i casi il flusso 

viene ripreso al primo 

comando che segue END 

IF

IF condition THEN

statement(s);

ELSE

statement(s);

END IF;

Esempio

CREATE OR REPLACE PROCEDURE Esempio IS

qty_on_hand NUMBER(5);

BEGIN

SELECT quantity INTO qty_on_hand

FROM inventory

WHERE product = 'TENNIS RACKET' FOR UPDATE OF 

quantity;

IF qty_on_hand > 0 THEN -- check quantity

UPDATE inventory SET quantity = quantity – 1

WHERE product = 'TENNIS RACKET';

INSERT INTO purchase_record

VALUES ('Tennis racket purchased', SYSDATE);

ELSE
INSERT INTO purchase_record

VALUES ('Out of tennis rackets', SYSDATE);

END IF;
COMMIT;

END;

IF-THEN-ELSIF

IF condition

TRUE

THEN actions

FALSE

ELSIF

condition

TRUE

THEN actions

FALSE

ELSE actions

 Simile a un comando 
SWITCH 

 Quando è possibile si usi 
ELSIF invece che un 
insieme di IF in cascata

 Il codice risultante sarà più 
leggibile

 Le condizioni devono 
essere mutualmente 
esclusive

 Dopo l’esecuzione il flusso 

viene ripreso al primo 

comando che segue END IF

IF-THEN-ELSIF Esempio

CREATE OR REPLACE PROCEDURE Esempio IS

v_title         article.title%TYPE;

v_length        article.length%TYPE;

v_descr         VARCHAR2(6);

BEGIN

SELECT title, length INTO v_title, v_length

FROM article

WHERE articlenum = &sv_articlenum;

IF v_length <=1500 THEN

v_descr := 'Brief';

ELSIF v_length BETWEEN 1501 and 2500 THEN

v_descr := 'Short';

ELSIF v_length BETWEEN 2501 and 4000 THEN

v_descr := 'Medium';

ELSE

v_descr := 'Long';

END IF;

DBMS_OUTPUT.PUT_LINE('Article ' || v_title || ' is ' || v_descr);

END;

Condizioni complesse

 I valori null sono gestiti tramite l’operatore IS NULL

 es: IF v_gender IS NULL THEN

 Qualsiasi espressione aritmetica che comprenda un NULL 

comporta il risultato NULL

 Nella concatenazione di più variabili la presenza di un NULL 

viene trattata come una stringa vuota

 Condizioni complesse vengono create utilizzando gli operatori 

logici NOT, AND, and OR
 es: IF v_length > 500 AND v_type = 'BUS' THEN ...

 es: IF v_length > 500 OR  v_type = 'BUS' THEN ...

 es: IF v_length > 500 OR  v_type = 'BUS'  AND 

v_type = 'LAW' THEN ...

 La precedenza tra gli operatori è così fissata: NOT, AND, OR

Logica a tre valori

NOT

TRUE

FALSE

NULL

FALSE

TRUE

NULL

TRUE

FALSE

NULL

OR TRUE FALSE NULL

TRUE

TRUE

TRUE

TRUETRUE

FALSE

NULL NULL

NULL

AND

TRUE

FALSE

NULL

TRUE FALSE NULL

TRUE

NULL NULL

NULL

FALSE FALSE

FALSE

FALSE

FALSE

 Le istruzioni di controllo del flusso gestiscono anche 

predicati in cui sono coinvolte variabili con valori NULL.

 Si può verificare esplicitamente se una variabile ha valore 

NULL mediante gli operatori IS [NOT] NULL

Esercizi in aula

 EX3: Scrivere una procedura che calcoli il valore totale degli ordini e indichi 

in output se tale valore è superiore o inferiore a 1000

 EX4: modificare il codice di EX3 in modo da ottenere il seguente output:

1. Se totale < 100 ‘Risultato scarso’

2. Se 100< totale < 1000 ‘Risultato in media’

3. Se 1000 < totale ‘Risultato buono’

Il concetto di eccezione I
 Cosa è una exception?

 Un identificatore PL/SQL che viene valorizzato durante l’esecuzione di un 

blocco

 L’esecuzione viene trasferita al corrispondente gestore dell’eccezione nella 

sezione exception del blocco

 Come avviene la valorizzazione?

 Automaticamente (implicitamente) quando si verifica un errore runtime

 Esplicitamente se nel codice è presente l’istruzione RAISE 

 Come vengono gestite?

 Includendo una routine corrispondente nella sezione exception

 Cosa avviene in caso contrario?

 Il blocco PL/SQL termina con un errore

 L’eccezione è propagata all’applicazione chiamante

 SQL*Plus mostra il corrispondente messaggio di errore

Il concetto di eccezione II

 Possono essere definiti molti tipi di eccezioni ognuno associato a un 

proprio insieme di comandi 

 Ogni gestore è identificato da una clausola WHEN, che specifica una o più 

eccezioni, seguita da un insieme di comandi

 Si può verificare una sola eccezione per volta

 Il gestore OTHERS

 Controlla ogni eccezione non trattata esplicitamente

 Deve essere l’ultima eccezione nella lista

 Le eccezioni possono essere:

 Internally defined: vengono attivate dal sistema automaticamente e sono 

associate a un codice di errore

 Predefined: vengono attivate dal sistema automaticamente e sono 

associate a un codice di errore e a un nome

 User-defined: sono definite e attivate da un utente tramite il comando 

RAISE

Gestione delle eccezioni

Individua l’eccezione

DECLARE

BEGIN

END;

L’eccezione 

si manifesta
EXCEPTION

L’eccezione 

viene gestita

Propaga l’eccezione

DECLARE

BEGIN

END;

L’eccezione 

si manifesta
EXCEPTION

L’eccezione non 

viene gestita

L’eccezione è 

propagata 

all’ambiente 

chiamante

Gestione delle eccezioni

EXCEPTION

WHEN exception1 [OR exception2 . . .] THEN

statement1;

statement2;

. . .

[WHEN exception3 [OR exception4 . . .] THEN

statement1;

statement2;

. . .]

[WHEN OTHERS THEN

statement1;

statement2;

. . .]

Eccezioni predefinite I

 PL/SQL predefinisce alcune eccezioni comuni

 Esempi: 

 NO_DATA_FOUND (ORA-01403)

• Una SELECT INTO ha restituito 0 righe

 TOO_MANY_ROWS (ORA-01422)

• Una SELECT INTO ha restituito più di una riga

 VALUE_ERROR (ORA-06502)

– Si è verificato un errore aritmetico, numerico, di conversione o su un 

vincolo

– es: si è tentato di assegnare il valore NULL a una variabile definita 

NOT NULL, oppure si è tentato di assegnare 9876 a una variabile 

definita NUMBER(2)

 ZERO_DIVIDE (ORA-01476)

 DUP_VAL_ON_INDEX (ORA-00001)

Eccezioni predefinite II

BEGIN

...

EXCEPTION

WHEN NO_DATA_FOUND THEN

statement1; 

statement2; 

WHEN TOO_MANY_ROWS THEN

statement1;

statement2; 

WHEN OTHERS THEN

statement1; 

statement2; 

END;

NO_DATA_FOUND

CREATE OR REPLACE PROCEDURE Esempio(v_writerid varchar2) IS

v_writername VARCHAR2(50);

BEGIN

SELECT fn || ' ' || ln INTO v_writername

FROM writer

WHERE writerid = v_writerid;

DBMS_OUTPUT.PUT_LINE('Writer ' || v_writerid || ' is ' 
|| v_writername);

EXCEPTION

WHEN NO_DATA_FOUND THEN

DBMS_OUTPUT.PUT_LINE('No such writer: '|| 
v_writerid);

END;

NO_DATA_FOUND

 Le funzioni di aggregazione SQL (es. AVG, SUM) restituiscono 

sempre un valore o NULL

 Un comando SELECT INTO che nella select list include solo funzioni 

di aggregazione non attiva mai l’eccezione NO_DATA_FOUND.

 Ovviamente ciò non è vero se il comando SELECT INTO prevede 

anche un raggruppamento

Esercizi in aula

 EX5: Scrivere una procedura che calcoli il valore totale degli ordini effettuati 

in una certa data. Nel caso in cui non sia presente nessun ordine viene 

visualizzato il messaggio “Nessun ordine presente per la data: ---------”

 EX6: Scrivere una procedura che permetta di visualizzare il nome del cliente 

relativo a una data fattura. Nel caso la fattura non sia presente visualizzare 

tramite il comando RAISE “La fattura --- non è stata registrata”

Cicli

 PL/SQL mette a disposizioni 4 istruzioni per il controllo 

dei cicli:

 Cicli semplici

 Ciclo WHILE

 Cicli FOR numerici

 Cicli FOR per cursori



53

Cicli semplici (Pre-Test)

 La condizione di uscita viene controllata prima 
dell’esecuzione di uno qualsiasi dei comand
 Se la condizione è verificata fin dall’inizio nessun

comando verrà eseguito

 Il solo comando EXIT determina l’uscita incondizionata 
dal ciclo più interno

CREATE OR REPLACE PROCEDURE Esempio(p_end_at NUMBER) IS

v_counter    NUMBER(2) := 1;

BEGIN

LOOP

EXIT WHEN v_counter > p_end_at;

DBMS_OUTPUT.PUT_LINE(v_counter);

v_counter := v_counter + 1;

END LOOP;

END;

Cicli semplici (Post-Test)

 La condizione di uscita viene controllata dopo la 
prima esecuzione dei comandi
 I comandi vengono eseguiti almeno una volta

CREATE OR REPLACE PROCEDURE Esempio(p_end_at NUMBER) IS

v_counter NUMBER(2) :=1;

BEGIN

LOOP

DBMS_OUTPUT.PUT_LINE(v_counter);

v_counter := v_counter + 1;

EXIT WHEN v_counter > p_end_at;

END LOOP;

END;

Cicli WHILE

 Ripete una sequenza di 

comandi finché la condizione è 

TRUE

 La condizione viene verificata 

prima di eseguire i comandi 

• Ciclo pre-test

 Il ciclo termina quando la 

condizione diviene FALSE o 

NULL

 Può essere utilizzato il 

comando EXIT per terminare in 

maniera anticipata il ciclo

WHILE condition

LOOP

statement1;

statement2;

. . .

END LOOP;

WHILE condition1

LOOP

statement1;

statement2;

EXIT WHEN condition2

. . .

END LOOP;

Esempio

CREATE PROCEDURE Esempio(p_end_at NUMBER) IS

v_counter     NUMBER(2) := 1;

BEGIN

WHILE v_counter <= p_end_at

LOOP

DBMS_OUTPUT.PUT_LINE(v_counter);

v_counter := v_counter + 1;

END LOOP;

END;

Cicli FOR numerici
 Un contatore implicito 

viene incrementato 
automaticamente ad 
ogni ciclo

 L’incremento è 
sempre di 1

 Il ciclo continua finché il 

contatore è < all’upper 

bound.

 Se il lower bound ha un 

valore superiore 

all’upper bound i 

comandi non vengono 

eseguiti

FOR counter IN [REVERSE] lower..upper

LOOP

statement1;

statement2;

. . .

END LOOP;

 lower e upper bound possono essere 
numeri, variabili, o espressioni che possano 
essere sempre valutati come interi

 Il counter è definito e può essere 
referenziato solo all’interno del ciclo

 Può essere utilizzato il comando EXIT per 
terminare in maniera anticipata il ciclo

Esempio

CREATE PROCEDURE Esempio(p_end_at NUMBER) IS

BEGIN

FOR v_count IN 1..p_end_at

LOOP

DBMS_OUTPUT.PUT_LINE(v_count);

END LOOP;

END;

Cicli Nested e Label

 I cicli FOR, WHILE, e simple possono essere 

innestati

 E’ possibile associare a un loop una label per 

semplificare la lettura del codice. La label potrà 

essere infatti inclusa dopo il comando END LOOP

 Per dichiarare una label vengono utilizzati i 

delimitatori (<<label>>)

Cicli Nested e Label

CREATE PROCEDURE Esempio IS

v_outercounter     NUMBER(2) := 1;

v_product          NUMBER(4);

BEGIN

<<Outer_loop>>

WHILE v_outercounter <= 3

LOOP

<<Inner_loop>>

FOR v_innercounter IN 1.. 5

LOOP

v_product := v_outercounter* v_innercounter;

DBMS_OUTPUT.PUT_LINE(v_outercounter || ' x ' ||

v_innercounter || ' = ' || v_product);

END LOOP Inner_loop;

DBMS_OUTPUT.PUT_LINE('------------------------');

v_outercounter := v_outercounter + 1;

END LOOP Outer_loop;

END;

Esercizi in aula

 EX7: Calcolare il valore totale delle fatture con codice da 1 a 5 se una fattura 

non è presente visualizzare un messaggio di errore tramite RAISE

Il concetto di cursore 
 Ogniqualvolta si sottoponga al sistema un comando SQL, Oracle alloca 

un’area di memoria in cui il comando viene analizzato ed eseguito. Tale 

area è detta context area.

 Un cursore è un puntatore alla locazione di memoria di una context area

 Ogni comando SQL eseguito da Oracle ha associato un proprio cursore

Cursor
context

area

active set

WriterID LN FN Phone 

L350 Lawton Pat  
W432 Waldeck Kristine (917) 361-8181 
C200 Cox Kelly (210) 783-5415 

C500 Cohen Julia Rice (910) 338-1875 

E235 Epstein Diane (610) 349-9689 

H400 Hall Valerie (710) 918-7767 

J525 Johnson Leroy W. (210) 895-2046 

 

Number of 

rows 

processed

Parsed 

command

statement

Due tipi di Cursori
 Cursori Impliciti

 Il server Oracle apre implicitamente un cursore durante 

l’esecuzione di un comando DML o di ogni query PL/SQL 

SELECT INTO

 Il cursore è gestito automaticamente

• Non si può utilizzare OPEN, FETCH, CLOSE per controllarlo

 PL/SQL fa riferimento al più recente cursore implico come 

cursore SQL

 Cursori Espliciti

 Sono dichiarati e maneggiati direttamente dal codice

 Sono utilizzati per processare le singole righe restituite da un 

comando SQL multiple-row

 Puntano alla riga corrente nell’ active set

Attributi dei cursori impliciti I
 E’ possibile utilizzare gli attributi del cursore sql per verificare 

il risultato di un comando SQL

SQL%ROWCOUNT Numero di righe coinvolte dal più
recente comando SQL

SQL%FOUND Attributo Boolean che è TRUE se 
l’ultimo comando SQL ha restituito 
almeno una riga

SQL%NOTFOUND Attributo Booleano che è TRUE se 
l’ultimo comando SQL non ha 
restituito nemmeno una riga

SQL%ISOPEN E’ sempre FALSE poiché PL/SQL 
chiude i cursori impliciti 
immediatamente dopo l’esecuzione

Attributi dei cursori impliciti  II

 Dai a ogni scrittore freelance un aumento del 25% e mostra 

il numero di righe modificate.

CREATE PROCEDURE Esempio IS

-- nessun cursore è dichiarato esplicitamente

BEGIN

UPDATE writer

SET amount = amount * 1.25

WHERE freelancer = 'Y';

DBMS_OUTPUT.PUT_LINE(SQL%ROWCOUNT ||' rows changed.');

COMMIT;

END;

SQL> /

6 rows changed.

PL/SQL procedure successfully completed.

Record PL/SQL

 Un record PL/SQL è un gruppo di attributi correlati memorizzati 
in una tabella, ognuno col proprio nome e tipo

 Un record PL/SQL è quindi un tipo composto in cui i singoli 
campi sono trattati come un’unità logica

 Sono convenienti per gestire le righe dell’ active set, poiché 
permettono di eseguire il FETCH di un’intera riga.

 I valori della riga vengono caricati nei campi corrispondenti

 I campi senza un valore iniziale sono inizializzati a NULL.

%ROWTYPE
 Dichiara una variabile di tipo record basandosi su un 

insieme di campi appartenenti a una tabella, vista o cursore.

 I campi nel record assumono il nome e il tipo di quelli nella 

tabella, vista o cursore.

 Ci si riferisce a un membro di un campo utilizzando la 

sintassi

 recordvariable_name.fieldname

 Il tipo e il numero delle colonne nel database può cambiare.

 E’ necessario anteporre a %ROWTYPE il nome della 

tabella, vista o cursore a cui il record è associato.



68

Creazione di record utilizzando 

%ROWTYPE
 Per dichiarare una variabile di tipo record al fine di memorizzare 

le informazioni presenti nella tabella ARTICLE

DECLARE

vr_article  article%ROWTYPE;

. . .

vr_article

vr_article.articlenum

vr_article.title

vr_article.issue

vr_article.length

vr_article.writerid

vr_article.type

Controllo di un cursore esplicito

Crea la 
context 

area

Definisce la 
query che 
produce i 

dati

DECLARE

Esegue la 
query

Identifica 
l’active set

Posiziona il 
cursore prima 

della riga 
iniziale

OPEN

Carica la riga 
corrente nelle 

variabili

Fa avanzare 
il cursore alla 
prossima riga

FETCH

Controlla 
l’esistenza di 
ulteriori righe

EMPTY?

Esegue 
nuovamente il 

FETCH se è 
stata trovata 

una riga

No

Rilascia 
l’active set

Disabilita il 
cursore e 
rilascia le 

risorse

CLOSE

Yes

Attributi dei Cursori espliciti
 Permettono di ottenere informazioni sui cursori espliciti

Attributo Tipo Descrizione

%ISOPEN Boolean Restituisce TRUE se il cursore è 

open

%NOTFOUND Boolean Restituisce TRUE se il FETCH più 

recente non ha restituito righe

%FOUND Boolean Restituisce TRUE se il FETCH più 

recente ha restituito righe.

%ROWCOUNT Number Restituisce il numero totale di 

righe restituite (ossia fetched)

Dichiarazione di un cursore 

esplicito

 select_statement è un qualsiasi comando SELECT

 Può includere join, operatori di set e subquery

 Se è necessario processare le righe in una determinata 
sequenza si può utilizzare la clausola ORDER BY nella query.

 E’ possibile fare riferimento a variabili all’interno della query, 
ma queste devono essere definite anticipatamente.

CURSOR cursor_name IS

select_statement;

Esempio

CREATE PROCEDURE Esempio IS

CURSOR writer_cursor IS

SELECT  ln, phone, amount

FROM    writer;

v_length NUMBER(4,0) := 1500;

CURSOR article_cursor IS

SELECT *

FROM   article

WHERE  length < v_length;

. . .

Apertura di un cursore

 Esegue l’interrogazione e identifica l’active set.

 Posiziona il puntatore prima della prima riga nell’active set.

 Le righe non vengono caricate nelle variabili fino all’esecuzione 

del comando FETCH

 Non si verifica alcuna eccezione se la query non restituisce 

valori.

OPEN cursor_name;

Leggere i dati dal cursore

 I dati possono essere inseriti in un record o in un insieme di 
variabili

 Dopo un FETCH, il cursore avanza alla prossima riga 
dell’active set

 Dopo ogni FETCH è necessario verificare se il cursore 
contiene delle righe

 Se un cursore non acquisisce valori l’active set è stato 
completamente elaborato

 Non vengono create delle eccezioni

 Le variabili/record mantengono i valori precedenti

FETCH cursor_name INTO [variable1, variable2, ...]

| record_name];



75

Caricamento dei dati dei cursori 

in variabili PL/SQL
CREATE PROCEDURE Esempio IS

CURSOR writer_cursor IS

SELECT ln, phone

FROM   writer

ORDER BY ln;

v_ln     writer.ln%TYPE;

v_phone  writer.phone%TYPE;

BEGIN

OPEN writer_cursor;

LOOP

FETCH writer_cursor INTO v_ln, v_phone;

EXIT WHEN writer_cursor%NOTFOUND;

DBMS_OUTPUT.PUT_LINE(RPAD(v_ln,40) || v_phone);

END LOOP;

CLOSE writer_cursor;

END;

Caricamento dei dati dei cursori 

in record PL/SQL
CREATE PROCEDURE Esempio IS

CURSOR writer_cursor IS

SELECT ln, phone

FROM   writer

ORDER BY ln;

vr_writer  writer_cursor%ROWTYPE;

BEGIN

OPEN writer_cursor;

LOOP

FETCH writer_cursor INTO vr_writer;

EXIT WHEN writer_cursor%NOTFOUND;

DBMS_OUTPUT.PUT_LINE(RPAD(vr_writer.ln,40) ||

vr_writer.phone);

END LOOP;

CLOSE writer_cursor;

END;

Chiusura di un Cursore

 Chiude il cursore dopo aver completato l’elaborazione.

 Disabilita il cursore rendendo indefinito l’active set.

 Non è possibile eseguire FETCH su un cursore chiuso.

 Provocherebbe una eccezione di tipo INVALID_CURSOR

 La riapertura del cursore provocherà la riesecuzione 

dell’interrogazione

CLOSE cursor_name;

 Semplifica l’utilizzo di cursori espliciti

 Il cursore è aperto e ne viene recuperata una riga per ogni iterazione; il 

cursore è chiuso automaticamente dopo l’elaborazione dell’ultima riga. 

 Il record PL/SQL che conterrà i dati viene definito automaticamente

 Le operazioni di OPEN, FETCH, e CLOSE avvengono 

automaticamente

Cicli FOR e cursori

FOR record_name IN cursor_name LOOP   

statement1;

statement2;

. . .

END LOOP;

ESEMPIO

CREATE PROCEDURE Esempio IS

CURSOR writer_cursor IS

SELECT ln, phone

FROM   writer

ORDER BY ln;

BEGIN

FOR vr_writer IN writer_cursor LOOP -- implicit open/fetch

DBMS_OUTPUT.PUT_LINE(RPAD(vr_writer.ln,40) ||  

vr_writer.phone);

END LOOP; -- Chiusura implicita

END;

 Recupera nome e cognome di ogni scrittore

 Si noti la riduzione nel numero dei comandi

 Nessuna dichiarazione per vr_writer

 Esecuzione automatica di OPEN, FETCH e CLOSE

Cursori con Parametri

 I parametri permettono di passare al cursore dei valori 

utilizzati nella query che carica i dati durante l’apertura.

 Un cursore può essere aperto più volte nello stesso blocco 

producendo active set diversi

 Non possono essere utilizzati assieme a un ciclo FOR 

poiché i parametri devono essere specificati durante 

l’operazione esplicita di OPEN

CURSOR cursor_name

[(parameter_name datatype, ...)]

IS

select_statement;

Cursori con Parametri

CREATE PROCEDURE Esempio IS

CURSOR writer_cursor (p_flstatus IN writer.freelancer%TYPE) IS

SELECT ln, phone

FROM writer

WHERE freelancer = p_flstatus; 

vr_writer  writer_cursor%ROWTYPE;

BEGIN

OPEN writer_cursor('Y');

LOOP

FETCH writer_cursor INTO vr_writer;

EXIT WHEN writer_cursor%NOTFOUND;

DBMS_OUTPUT.PUT_LINE(RPAD(vr_writer.ln,40) ||

vr_writer.phone);

END LOOP;

CLOSE writer_cursor;

END;

Esercizi in aula

 EX8: Creare un cursore che restituisca separatamente l’importo di tutte le 

fatture

 EX9: Creare un cursore che restituisca separatamente la somma degli 

importi delle fatture di importo minore e maggiore di mille.

FOR UPDATE

 Applica un lock alle righe selezionate dal cursore in modo che sia 
possibile modificare o cancellare i valori all’interno del codice

 Il lock è applicato al momento dell’apertura del cursore non durante 
la fase di fetch

 Il lock è rilasciato al momento del COMMIT o ROLLBACK da 
eseguire al termine del ciclo

 L’esecuzione di COMMIT o ROLLBACK per ogni riga 
provoca errore (ORA-01002)

 Se il cursore applica una selezione su più tabelle tramite FOR 
UPDATE è possibile limitare il lock a una sola tabella. Il lock è 
applicato solo alle righe delle tabelle di cui è citato un campo nella 
clausola FOR UPDATE.

 La clausola FOR UPDATE è l’ultima di ogni query di SELECT.

SELECT ... 

FROM ...

FOR UPDATE [OF column_reference][NOWAIT];

FOR UPDATE

 NOWAIT indica al server di non attendere se sulle tabelle è 

attivo un lock di un’altra sessione.

 Si verifica una exception

 Il controllo è restituito al programma che può eseguire altre 

operazioni prima di tentare di riacquisire il lock

SELECT ... 

FROM ...

FOR UPDATE [OF column_reference][NOWAIT];

FOR UPDATE Esempio

CREATE PROCEDURE Esempio IS

CURSOR c_stud_zip IS

SELECT s.student_id, z.city

FROM student s, zipcode z

WHERE z.city = 'Brooklyn'

AND   s.zip = z.zip

FOR UPDATE OF phone;

BEGIN

FOR r_stud_zip IN c_stud_zip

LOOP

UPDATE student

SET phone = '333'|| substr(phone,4)

WHERE student_id = r_stud_zip.student_id;

END LOOP;

COMMIT;

END;

Cosa viene bloccato?

Quando?

Il COMMIT è eseguito alla fine

WHERE CURRENT OF

 Referenzia la riga corrente di un cursore esplicito.

 Permette di eseguire UPDATE o DELETE della riga corrente 

utilizzando una clausola WHERE semplificata.

 Non richiede di creare la condizione che specifichi a quale riga 

applicare l’operazione poiché questa viene applicata alla riga corrente.

 E’ necessario utilizzare FOR UPDATE nella definizione del 

cursore in modo da applicare un lock sulla tabella

 In caso contrario si verificherà un errore

WHERE CURRENT OF cursor ;

WHERE CURRENT OF

CREATE PROCEDURE Esempio IS

CURSOR c_stud_zip IS

SELECT s.student_id, z.city

FROM student s, zipcode z

WHERE z.city = 'Brooklyn'

AND   s.zip = z.zip

FOR UPDATE OF phone;

BEGIN

FOR r_stud_zip IN c_stud_zip

LOOP

UPDATE student

SET phone = '718'|| substr(phone,4)

WHERE CURRENT OF c_stud_zip;

END LOOP;

COMMIT;

END;

Esercizi in aula

 EX10: Alzare del 10% il prezzo di tutti i prodotti forniti dal fornitore ‘xxx’

Procedurale vs Dichiarativo
 La modalità di calcolo da preferire è quella che massimizza le 

prestazioni (purché non complichi eccessivamente il codice)

 La principale regola di massima prevede che sia demandata 

all’ottimizzatore la modalità di accesso ai dati

 E’ meglio far eseguire al sistema una query complessa piuttosto 

che molte query semplici

 Una valutazione più approfondita richiede di conoscere le 

modalità di accesso e di ottimizzazione utilizzate dal DBMS… E’ 

anche per questo motivo che le studieremo

Procedurale vs Dichiarativo

 Un esempio: restituire in output separatamente l’importo di tutte le fatture con 

codice compreso tra 1 e 5

CURSOR cursore_importi IS

SELECT D_NUMF,sum(D_QTA*D_PREZZO) as IMPORTO

FROM   dettaglio

WHERE D_NUMF BETWEEN 1 AND 5

GROUP BY D_NUMF;

…

LOOP 

FETCH cursore_importi into vr_importi;

EXIT WHEN cursore_importi%NOTFOUND;

DBMS_OUTPUT.PUT_LINE('La fattura: ' || vr_importi.D_NUMF

|| ' e'' di importo: ' || vr_importi.IMPORTO);

END LOOP;

Procedurale vs Dichiarativo

 Un esempio: restituire in output separatamente l’importo di tutte le fatture con 

codice compreso tra 1 e 5

 Meno efficiente scandisce il database 5 volte (in assenza di indici)

…

FOR counter IN 1.. 5 LOOP

SELECT sum(D_QTA*D_PREZZO) as IMPORTO INTO v_importo

FROM   dettaglio

WHERE D_NUMF = counter

GROUP BY D_NUMF;

DBMS_OUTPUT.PUT_LINE('La fattura: ' || counter

|| ' e'' di importo: ' || v_importo);

END LOOP;

Procedurale vs Dichiarativo

 Un esempio: restituire in output l’importo totale delle fatture che hanno 

singolarmente un importo > 1000 e <= 1000

CURSOR cursore_importi IS

SELECT D_NUMF,sum(D_QTA*D_PREZZO) as IMPORTO

FROM   dettaglio

group by D_NUMF;

…

open cursore_importi;

LOOP 

FETCH cursore_importi into vr_importi;

EXIT WHEN cursore_importi%NOTFOUND;

if vr_importi.IMPORTO < 1000 then

v_TotSmall := v_TotSmall + vr_importi.IMPORTO;

else

v_TotBig := v_TotBig + vr_importi.IMPORTO;

end if;

END LOOP;

Procedurale vs Dichiarativo

 Un esempio: restituire in output l’importo totale delle fatture che hanno 

singolarmente un importo > 1000 e <= 1000

 Meno efficiente scandisce 2 volte il database: il calcolo della clausola 

having non può sfruttare strutture a indice

CURSOR cursore_importi IS

SELECT SUM(IMPORTO) INTO v_TotBig

FROM (SELECT D_NUMF,sum(D_QTA*D_PREZZO) as IMPORTO

FROM   dettaglio

GROUP BY D_NUMF

HAVING IMPORTO > 1000);

…

SELECT SUM(IMPORTO) INTO v_TotSmall

FROM (SELECT D_NUMF,sum(D_QTA*D_PREZZO) as IMPORTO

FROM   dettaglio

GROUP BY D_NUMF

HAVING IMPORTO <= 1000);

Sommario
 Tipi di cursore:

 Impliciti: Utilizzati in tutti i comandi DML e per le query single-

row.

 Espliciti: Utilizzabili per le query a 0,1 o più righe.

 I cursori espliciti devono essere gestiti dal codice

 DECLARE

 OPEN

 FETCH

 CLOSE

 Lo stato del cursore può essere valutato utilizzando i suoi 

attributi

Esercizi in aula

 EX11: Scrivere una funzione che verifichi se un certo prodotto p è presente 

in quantità > q

 EX12: Scrivere una procedura che emetta un ordine per ogni prodotto 

presente in quantità < 100;

Esercizi sui trigger

 EX13: Scrivere  trigger che all’inserimento di un nuovo dettaglio d’ordine 

aggiorni la quantità disponibile per il prodotto

 EX14: Scrivere un trigger che prima di accettare un ordine verifichi che il 

cliente non abbia superato il valore massimo degli acquisti, ossia che la 

somma degli importi acquistati e non pagati non superi 10000. In tal caso 

l’ordine non deve essere inserito e deve essere visualizzato il messaggio 

“Superato il credito massimo”


